

Guidelines for use of library Keypad_MC16

G. D. (Joe) Young – Feb 2013

The Arduino user quickly finds that the number of pins available to connect to external devices the user wishes to control is limited. A 12- or 16-key keypad uses 7 or 8 lines (although these same lines can sometimes be shared—Keypad library, liquid crystal display). Using keypads connected to the inter-ic (IIC or I2C) bus originally developed by Phillips (now NXP) uses only two arduino pins and ground and the I2C bus can have many additional devices also simultaneously connected to the same two lines.

The Keypad_MC16 library extends the Keypad library so that a keypad requiring 16 lines or fewer connected to a Microchip MCP23016 I2C parallel-port behaves very nearly the same as one directly-connected to the precious arduino pins. Moreover, two or more such connections can operate simultaneously to permit applications needing more than 64 keys.


Hardware

For a complete description of the MCP23016, see the file MCP23016.pdf.

For the wiring diagram for an interface to an example keypad, see the file usingMCP2016.jpg.


These files and this one will be found in the docs folder inside the Keypad_MC16 library folder.

Shown here is a photo of a breadboard hookup. Note that it is almost irrelevant how the connections to the keypad are made—they are specified in the software. But it may simplify initial understanding if the row pins are connected to MCP23016 pins 20, 21, 22, and 23 and column pins to 24, 25, and 26 (bits gp0.0, gp0.1..gp0.6)—as assumed in the HelloKeypad_MC16 example sketch. Also note the necessary pullup resistors. The two keypads are connected as a 4 row X 7 column keypad (HelloKpd28_MC16).



Keypad layout

The software assumes that the key pad has the arrangement of switches shown in the adjacent diagram—that is, each key when pressed joins a unique pair of row and column pins. For example, pushing the 9 key joins column 2 and row 2. The external connections to the keypad are made to the row pins and the column pins. A common labeling of the switches is shown (the “telephone” keypad layout).



Software

The Keypad_MC16 library uses the Keypad library and Wire library. The Wire library will be already available in any arduino installation. Place the Keypad folder and the Keypad_MC16 folder in the 'libraries' folder within your sketchbook folder. You may need to restart the arduino environment for the additions to show up in the sketchbook/libraries menu.

For a quick start, select the example sketch HelloKeypad_MC16 from the Keypad_MC16/Examples menu, upload, and then serial monitor. Keys pressed on the keypad should show up on the serial output. A detailed description of the various parts of this sketch follows.

```
#include <Keypad_MC16.h>
#include <Keypad.h>
#include <Wire.h>
```

/*The first three lines make the functions of the Keypad_MC16 library available to this sketch. All three of them are necessary. Keypad.h has the principal functionality, Keypad_MC16.h extends Keypad's input/output operations to make them use I2C, and Keypad_MC16 in turn calls on the Wire library to actually perform the communication between the arduino and the MCP23016.

*/

```
const byte ROWS = 4; //four rows
const byte COLS = 3; //three columns
```

/*Defines the size of the keypad.

*/

```
char keys[ROWS][COLS] = {
  {'1','2','3'},
  {'4','5','6'},
  {'7','8','9'},
  {'*','0','#'}
};
```

/*Defines the layout of the keypad, and what character is produced by the library when the corresponding key is pushed. For instance, when the key in the second row and third column is pushed, the character '6' is produced. This matrix can be defined with any single characters desired.

*/

```
byte rowPins[ROWS] = {0, 1, 2, 3}; //connect to the row pinouts of the keypad
byte colPins[COLS] = {4, 5, 6}; //connect to the column pinouts of the keypad
```

```
/*Defines how the keypad wires are connected to the bit numbers of the MCP23016. Bit 0 comes out on the IC's pin 20, bit 1 on pin 21, ..., bit 4 on pin 24, ..., bit 7 on pin 27.
```

```
*/
```

```
int i2caddress = 0x20;
```

```
/* Specify the address on the I2C bus that this particular IC responds to. The MCP23016 can have hexadecimal addresses 0x20, 0x21, .., 0x27 determined by the connections to the 3 address lines on pins 16, 17, and 18.
```

I2C addresses can be somewhat confusing because the least significant bit of the byte going out on the I2C bus that contains the address is the read/write bit . Consequently, addresses are sometimes referred to by the whole 8-bits, and sometimes by just the 7 most significant bits as if they were right-justified—the arduino Wire library uses this second convention.

```
*/
```

```
Keypad_MC16 kpd = Keypad_MC16( makeKeymap(keys), rowPins, colPins, ROWS, COLS, i2caddress );
```

```
/*This line calls the Keypad_MC16 library's constructor with all of the specifications outlined above to create the object kpd. The first four arguments are the same as for the direct-connection Keypad's constructor, the last argument is the I2C address for this keypad's I2C integrated circuit.
```

```
*/
```

```
void setup(){  
  Serial.begin(9600);  
  kpd.begin();  
}
```

```
/*Starts the serial monitor library, specifying the baud rate to be used, and starts the Keypad_MC16 library. kpd.begin( ) is required to make the library aware of the state of the I2C port the library will use, as well as starting the Wire library..
```

```
*/
```

```
void loop(){  
  char key = kpd.getKey();  
  
  if (key){  
    Serial.println(key);  
  }  
} // end of HelloKeypad_I2C sketch
```

```
/*The loop repeatedly scans the keypad—kpd.getKey( )—and if a keypress is discovered, the corresponding character is sent to the serial monitor—Serial.println(key).
```

```
*/
```

Comparison with HelloKeypad

The MC16 version of the keypad library is very similar to the direct-connection Keypad library. Consequently, if you have already got a sketch going which uses the Keypad library, it can be switched to using the Keypad_MC16 library with a minimum of changes:

- Add the two include statements: `#include <Keypad_MC16.h>` and `#include <Wire.h>`
- Modify the pin assignments in arrays `rowPins[ROWS]` and `colPins[COLS]` to conform to the wiring between the keypad row and column pins and the MCP23016
- Change the constructor statement to `Keypad_MC16` and add the I2C address parameter, keeping the object name that is in your already-working sketch. (Here, it's `kpd`.)
- Add the `kpd.begin()` statement in `setup()`

Other Features

The Keypad_MC16 library has four functions which will enable some sharing of the I2C port between a keypad that doesn't use all 16 of the MCP23016's i/o pins and other digital i/o. For example, if the keypad is a commonly available 12-key version such as is described for the HelloKeypad_MC16 example, then there are 9 extra pins available on the port chip that could be used—say for driving LED indicator circuits, or whatever. But keep in mind the drive capability of the MCP23016—it's comparable to the arduino's digital output pin drivers. The high output can source 25 mA, the low output can sink 25mA. Inputs do not have pullups internally, so external pullups are required for keypad pins (and maybe for pins used for other uses).

Because the I2C chip's registers must be written all at once, these functions are needed so that the operation of the keypad can be kept separated from the operation of the spare pin(s).

`iodir_read()` (no parameters, returns a word) is called to get the current state of all 16 bits of the MCP23016's IODIR0, 1 registers which defines the port pins as inputs or outputs at each bit position.

`iodir_write(value)` writes all 16 direction-defining bits to the IODIR0, 1 register pair. Each of the 16 bits of the word `value` determines if the corresponding pin is input (bit = 1) or output (bit = 0).

`pinState_set()` (no parameters, returns a word) is called to get the current state of all 16 bits of the port, and `port_write(value)` writes all 16 bits of `value` to the port and returns nothing.

The following example of a function to toggle the spare pin (bit 7) illustrates the use of these four Keypad_MC16 member functions (first line setting bit 7 as output goes in `setup()` section):

```
kpd.iodir_write( kpd.iodir_read() & 0xff7f ); // gp0 msb output for led

word portState; // to hold local copy of kpd port
const word ledPin = 0x0080; // msbit gp0 to be used for LED drive

void toggleLEDpin( void ) {
    portState = kpd.pinState_set(); // get current kpd port state into portState
    if( portState & ledPin ) {
        portState &= ~ledPin;
    } else {
        portState |= ledPin;
    }
    kpd.port_write( portState ); // write modified state to port
} // toggleLEDpin()
```